

Exemples de requêtes dans data.bnf.fr

1.1 Découvrir le modèle de données

Quelles sont les propriétés pour décrire la ressource <http://data.bnf.fr/ark:/12148/cb11933798p>

```
SELECT DISTINCT ?p ?o WHERE {
<http://data.bnf.fr/ark:/12148/cb11933798p> ?p ?o.
}
```

1.2 Informations sur une personne ou une organisation

1.2.1 Informations sur l'auteur

Exemple : Toutes les dates biographiques des auteurs (anniversaire et année de naissance et de mort) ainsi que leur nom,

```
SELECT ?auteur ?jour ?date1 ?date2 ?nom where {
?auteur foaf:birthday ?jour.
?auteur bio:birth ?date1.
?auteur bio:death ?date2.
OPTIONAL {?auteur foaf:name ?nom}
}
ORDER BY (?jour)
```

Exemple 2 : toutes les informations sur Victor Hugo et toutes les formes du nom de cette personne

```
SELECT DISTINCT ?nom_complet ?nom ?prenom ?forme_retenue
?formes_rejetees ?pays ?langue ?sexe ?anniversaire ?date_naissance
?lieu_naissance ?date_mort ?lieu_mort ?periode_activite ?domaine_activite
?bio WHERE {
<http://data.bnf.fr/ark:/12148/cb11907966z#foaf:Person> foaf:gender ?sexe;
<http://rdvocab.info/ElementsGr2/countryAssociatedWithThePerson>
?pays;
<http://rdvocab.info/ElementsGr2/languageOfThePerson> ?langue;
<http://rdvocab.info/ElementsGr2/dateOfBirth> ?date_naissance;
<http://rdvocab.info/ElementsGr2/placeOfBirth> ?lieu_naissance;
foaf:birthday ?anniversaire;
<http://rdvocab.info/ElementsGr2/dateOfDeath> ?date_mort;
<http://rdvocab.info/ElementsGr2/placeOfDeath> ?lieu_mort;
<http://rdvocab.info/ElementsGr2/fieldOfActivityOfThePerson>
?domaine_activite;
<http://rdvocab.info/ElementsGr2/biographicalInformation> ?bio;
foaf:name ?nom_complet;
foaf:familyName ?nom;
foaf:givenName ?prenom.
<http://data.bnf.fr/ark:/12148/cb11907966z> skos:altLabel ?formes_rejetees;
skos:prefLabel ?forme_retenue
OPTIONAL {<http://data.bnf.fr/ark:/12148/cb11907966z#foaf:Person>
<http://rdvocab.info/ElementsGr2/periodOfActivityOfThePerson>
?periode_activite}}
```



Exemple 3 : L'anniversaire des compositeurs, avec leurs noms et leurs années de naissance et de mort

```
SELECT ?auteur ?jour ?date1 ?date2 ?nom where {
?auteur foaf:birthday ?jour.
?doc bnfroles:r220 ?auteur.
?auteur <http://rdvocab.info/ElementsGr2/dateOfBirth> ?date1.
?auteur <http://rdvocab.info/ElementsGr2/dateOfDeath> ?date2.
OPTIONAL {?auteur foaf:name ?nom}
}
ORDER BY (?jour)
```

Exemple 4 : retrouver, pour chaque auteur, les pages comprenant les documents au sujet de cet auteur

```
SELECT DISTINCT ?auteur ?documents_a_propos_de WHERE {
?doc dcterms:creator ?uri_auteur .
?documents_a_propos_de dcterms:subject ?uri_auteur .
?uri_auteur foaf:name ?auteur
FILTER REGEX (?documents_a_propos_de, ".*studies.*")
}
```

```
SELECT DISTINCT ?auteur ?documents_a_propos_de WHERE {
?doc dcterms:creator ?uri_auteur .
?documents_a_propos_de dcterms:subject ?uri_auteur .
?uri_auteur foaf:name ?auteur
FILTER (!REGEX (?documents_a_propos_de, "http://data.bnf.fr/.*))
}
```

**Cette deuxième requête renvoie uniquement les pages extérieures à data.bnf.fr*

Exemple 5 : retrouver, pour un auteur (ici Charles Baudelaire), les documents au sujet de cet auteur

```
SELECT DISTINCT ?document WHERE {
?document dcterms:subject <http://data.bnf.fr/ark:/12148/cb118905823> .
}
```

1.2.2 Liens entre auteurs et œuvres

Exemple : tous les documents de Victor Hugo comme auteur du texte, et, le cas échéant, les œuvres attachées

```
SELECT DISTINCT ?doc ?oeuvre ?titre WHERE
{?doc marcrel:aut <http://data.bnf.fr/ark:/12148/cb11907966z#foaf:Person>. Optional {?oeuvre
rdarerelationships:expressionOfWork ?doc ; rdfs:label ?titre.}}
```

*attention, il n'existe pas d'œuvres pour tous les documents

<http://data.bnf.fr/semanticweb/#Ancre3>

1.3 Information sur une œuvre



Exemple 1 : Retrouver tous les titres du Roman de la Rose et la note associée

```
SELECT DISTINCT ?titre_forme_internationale_francais ?formes_rejetees ?note_associee
WHERE {
<http://data.bnf.fr/ark:/12148/cb166125510> skos:altLabel ?formes_rejetees;
skos:prefLabel ?titre_forme_internationale_francais;
skos:editorialNote ?note_associee
}
```

Exemple 2 : retrouver la liste des éditions d'une même œuvre (ici, les Fleurs du mal)

```
SELECT DISTINCT ?edition ?title ?date ?editeur WHERE {
<http://data.bnf.fr/ark:/12148/cb11947965f> foaf:focus ?Oeuvre .
?edition rdarelationships:workManifested ?Oeuvre.

OPTIONAL {?edition dcterms:date ?date}
OPTIONAL {?edition dcterms:title ?title}
OPTIONAL {?edition dcterms:publisher ?editeur}
}
```

1.4 Liens entre auteurs et expressions : fonctions d'un auteur sur des documents

Exemple 1 : Trouver les traducteurs d'ouvrages italiens en français

```
SELECT DISTINCT ?notice ?name WHERE {
?Work dcterms:language <http://id.loc.gov/vocabulary/iso639-2/ita> .
?Exp bnfroles:r680 ?traducteur ;
dcterms:language <http://id.loc.gov/vocabulary/iso639-2/fre>.
?Work rdarelationships:expressionOfWork ?Exp .
?traducteur foaf:name ?name.
?notice foaf:focus ?traducteur.
}
```

Exemple 2 : Retrouver le nom et prénom de tous les photographes

```
SELECT DISTINCT ?Photographe ?Prenom ?Nom WHERE
{
?expression bnfroles:r530 ?Photographe .
?Photographe a foaf:Person.
OPTIONAL {?Photographe foaf:givenName ?Prenom.}
OPTIONAL {?Photographe foaf:familyName ?Nom.}
}
```

1.5 Expositions virtuelles

Expositions virtuelles de la BnF et leurs descriptions



```
SELECT ?exposition ?titre ?sujet
WHERE {?exposition a <http://data.bnf.fr/ontology/bnf-onto/expositionVirtuelle> ; dcterms:title
?titre ; dcterms:subject ?sujet.}
```

1.6 Sujets (RAMEAU)

Exemple 1 : Pour un sujet Rameau, récupérer tous les termes spécifiques de niveau 1 et, le cas échéant, de niveau 2

```
SELECT DISTINCT ?original_rameau ?prefLabel ?uri_a ?label_a ?uri_b ?label_b
WHERE {
?original_rameau skos:prefLabel ?prefLabel ;
    skos:narrower ?uri_a .
MINUS {?original_rameau foaf:focus ?focus .}
?uri_a skos:prefLabel ?label_a .
OPTIONAL {
?uri_a skos:narrower ?uri_b .
?uri_b skos:prefLabel ?label_b .
}
}
```

Exemple 2 : documents au sujet du thème « Escrime » (URI <http://data.bnf.fr/ark:/12148/cb11931273g>)

```
SELECT ?doc {
?doc dcterms:subject <http://data.bnf.fr/ark:/12148/cb11931273g>.
}
```

Exemple : Les éditions qui ont pour sujet la photographie

```
SELECT DISTINCT ?Edition ?title WHERE {
?Edition a frbr-rda:Manifestation;
    dcterms:subject <http://data.bnf.fr/ark:/12148/cb11933113t>;
    dcterms:title ?title.
}
```

1.7 Identifiants

Exemple 1 : Pour un ISNI donné, récupérer le nom et prénom de la personne concernée (ici Alexandre Dumas)

```
SELECT DISTINCT ?nom ?prenom WHERE
{
?person isni:identifieurValid "0000000121012885" ;
    foaf:focus ?identity.
?identity foaf:familyName ?nom;
    foaf:givenName ?prenom.
}
```



Exemple 2 : Trouver l'identifiant ARK d'une notice à partir de son numéro FRBNF (ici, œuvre « les travailleurs de la mer »)

```
SELECT DISTINCT ?idArk WHERE
{
?idArk bnf-onto:FRBNF "11992081"^^xsd:integer.}
```

Exemple 3 : Retrouver la ou les œuvres correspondant à un ISBN avec le nom de l'auteur

Propriété utilisée : bnf-onto:isbn

Exemple : Trouver « Extension du domaine de la lutte » à partir de l'ISBN 2-7028-4777-3.

```
SELECT DISTINCT ?work ?title ?name WHERE
{
?work rdfs:label ?title;
  dcterms:creator ?creator.
?work rdarelationshps:expressionOfWork ?expression.
?manifestation rdarelationshps:expressionManifested ?expression.
?manifestation bnf-onto:isbn "2-7028-4777-3".
?creator foaf:name ?name.}
```

1.8 Vocabulaires et référentiels : genres musicaux, langue, pays, type de document, ouvrages jeunesse

Exemple 1 : sélectionner tous les documents de type image

```
SELECT ?image where {
?image dc:type <http://purl.org/dc/dcmitype/StillImage>.
} Limit 100
```

Exemple 2 : Lister les œuvres musicales par genre, en commençant par les genres les plus représentés

```
SELECT DISTINCT ?genre ?label COUNT (?work) AS ?nbWork
WHERE {
?work mo:genre ?genre.
?genre skos:prefLabel ?label }
ORDER BY DESC (?nbWork)
```

Exemple 3 : Sélectionner tous les ouvrages adaptés pour la jeunesse et l'œuvre adaptée correspondante

```
SELECT ?uri ?oeuvre WHERE {
?manifestation bnf-onto:ouvrageJeunesse "true"^^xsd:boolean ;
rdarelationshps:workManifested ?oeuvre ;
  rdfs:seeAlso ?uri.}
```



1.9 Requête sur des dates

Jour de naissance (anniversaire) sous la forme mm-jj : foaf:birthday

La date exacte de naissance ou de mort sous la forme aaaa-mm-jj : bio:Birth et bio:Death

Les années de naissance et de mort uniquement : bnf-onto:firstYear et bnf-onto:last year

Exemple 1 : liste des auteurs morts avant 1924

```
SELECT distinct ?nom ?auteur ?mort WHERE {  
  ?oeuvre dcterms:creator ?auteur.  
  ?auteur bio:death ?mort ; foaf:name ?nom.  
  FILTER (xsd:integer (?mort) < "1924"^^xsd:integer )  
}  
ORDER BY DESC (?mort)
```

*ne ramène pas les dates qui ne correspondent pas à un nombre entier (e.g. les dates incertaines n'apparaissent pas ici

Exemple 2 : liste des auteurs nés avant 1500, triés par date de naissance

```
SELECT ?auteur ?naissance where {?auteur bnf-onto:firstYear ?naissance.  
?auteur a foaf:Person.  
Filter (?naissance < "1500"^^xsd:integer)  
}  
ORDER BY ASC (?naissance)
```

1.10 Lieux, notices géographiques

Exemple 1 : tous les documents numérisés au sujet des notices géographiques

```
SELECT DISTINCT ?lieu ?doc ?docnum where {  
  ?lieu rdf:type geo:SpatialThing.  
  ?doc dcterms:subject ?concept.  
  ?concept foaf:focus ?lieu.  
  ?doc rdarerelationships:electronicReproduction ?docnum.  
}
```

Exemple 2 : tous les documents numérisés au sujet des notices géographiques et des thèmes associés

```
SELECT DISTINCT ?lieu ?doc ?docnum where {  
  {  
    ?lieu rdf:type <http://www.w3.org/2003/01/geo/wgs84_pos#SpatialThing>.  
    ?doc dcterms:subject ?concept.  
    ?concept foaf:focus ?lieu.  
    ?doc rdarerelationships:electronicReproduction ?docnum.}  
  UNION {  
    ?doc dcterms:subject ?rameau.  
    ?lieu skos:exactMatch ?rameau.}  
}
```



Exemple 3 : les lieux alignés avec leurs thèmes correspondants

```
SELECT ?lieu ?concept where {  
?c a <http://www.w3.org/2003/01/geo/wgs84_pos#SpatialThing>. ?lieu foaf:focus ?c. ?lieu  
skos:closeMatch ?concept.}
```

1.11 Spectacles

Exemple : dates, titres lieux et participants des spectacles dans data.bnf.fr

```
SELECT distinct ?spectacle ?nom ?lieu ?date ?participants WHERE {?spectacle a  
<http://rdvocab.info/uri/schema/FRBREntitiesRDA/Work>; a  
<http://purl.org/dc/dcmitype/Event> ; rdfs:label ?nom ; dcterms:date ?date ; dcterms:contributor  
?participants ; schemaorg:location ?lieu.} Limit 100
```

1.12 Périodiques

Exemple : périodiques numérisés dans Gallica et leurs ISSN

```
SELECT ?periodique ?titre ?issn ?Gallica WHERE {?periodique  
rdarelations:electronicReproduction ?Gallica ; bibo:issn ?issn ; rdfs:label ?titre.}
```

1.13 Liens vers des documents numérisés dans Gallica

Exemple : les photographies* numérisées dans Gallica, avec leur URL et leur titre

```
SELECT DISTINCT ?Doc ?title ?URLGallica WHERE {  
?Expression bnfroles:r530 ?Creator.  
?Doc rdarelations:expressionManifested ?Expression.  
?Doc rdarelations:electronicReproduction ?URLGallica; dcterms:title ?title.  
}
```

*Documents liés à une personne ou organisation avec le rôle « photographe »

Exemple 2 : tous les portraits d'auteurs dans Gallica

```
SELECT distinct ?auteur ?url WHERE {  
?auteur rdf:type skos:Concept ; foaf:focus ?person.  
?doc rdarelations:electronicReproduction ?url ; dcterms:subject ?auteur ;  
dcterms:subject <http://data.bnf.fr/ark:/12148/cb11932843f>.}
```